


*Collège du Sud, Bulle*

## Applications des mathématiques

# Simulation



*Version pour Mathematica*  
*Edition 2009/2010*  
*Marcel Déléze, Eugène Pasquier*  
[www.collegedusud.ch/applmaths/](http://www.collegedusud.ch/applmaths/)

# § 1 Nombres pseudo-aléatoires à distribution spécifiée

## § 1.1 Introduction

Pour résoudre un problème dans lequel intervient le hasard, il existe trois familles de méthodes:

### ■ Méthode théorique

Le calcul des probabilités sera étudié plus tard, en 4<sup>ème</sup> année, dans le cadre du cours de mathématiques. Nous nous contentons ici d'en faire une simple mention.

Lorsque le problème est assez simple, il peut être résolu d'une manière exacte et complète par le calcul des probabilités. Mais, pour des problèmes plus complexes, il est généralement difficile, voire même impossible, de donner une solution explicite. Il est utile de connaître d'autres approches.

### ■ Méthode expérimentale

Il est parfois possible d'organiser une série d'expériences. Par exemple, lancer un dé et noter les résultats obtenus est facile. Mais, pour obtenir une certaine précision, il faut faire un grand nombre d'expériences ce qui est coûteux.

### ■ Simulation (ou méthode de Monte Carlo)

Il existe des algorithmes capables de produire des "nombres pseudo-aléatoires" c'est-à-dire des nombres qu'on peut utiliser à la place des nombres donnés par le dé. L'ordinateur étant capable d'effectuer un grand nombre d'expériences en peu de temps, on peut obtenir des résultats plus précis et à meilleur compte qu'avec la méthode d'expérimentation.

Alors que les nombres aléatoires sont engendrés par un phénomène aléatoire (par exemple par tirage à la roulette), les nombres pseudo-aléatoires sont calculés au moyen d'un algorithme; ils offrent néanmoins les mêmes caractéristiques que les nombres aléatoires et peuvent avantageusement les remplacer dans les calculs.

On appelle "méthode de Monte-Carlo" toute méthode de calcul utilisant des nombres pseudo-aléatoires. Cette dénomination se réfère à la roulette du célèbre casino de Monte-Carlo. Parmi les applications des méthodes de Monte-Carlo, citons

- \* la programmation de jeux de hasard (dés, cartes, lotto, roulette, ...);
- \* la simulation de phénomènes aléatoires (
  - cheminements d'un neutron au sein d'une matière fissile,
  - fluctuations de charge d'un central téléphonique,
  - longueur de la file d'attente à un guichet, ...).

## § 1.2 Nombres pseudo-aléatoires à distribution discrète

### ■ Distribution de Bernoulli

```
n = 40;
```

```
? BernoulliDistribution
```

```
BernoulliDistribution[p] represents a Bernoulli distribution with probability parameter p.  
>>
```

Par exemple, une expérience aléatoire consistant à lancer un dé, le nombre de six est

```
six = BernoulliDistribution[ $\frac{1}{6}$ ];
RandomInteger[six, n]
{0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 1, 0, 0, 1, 0, 0, 0, 0,
 0, 1, 1, 0, 0, 0, 1, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

## ■ Distribution binomiale

?BinomialDistribution

BinomialDistribution[ $n, p$ ] represents a  
binomial distribution with  $n$  trials and success probability  $p$ . »

Par exemple, une expérience aléatoire consistant à lancer 10 dés, le nombre de six est

```
six10 = BinomialDistribution[10,  $\frac{1}{6}$ ];
RandomInteger[six10, n]
{2, 1, 1, 2, 3, 2, 0, 1, 0, 2, 1, 2, 0, 2, 1, 2, 2, 2,
 4, 3, 1, 4, 0, 1, 2, 1, 4, 3, 1, 1, 4, 4, 1, 1, 2, 0, 3, 1, 2, 0}
```

## ■ Distribution discrète uniforme

?DiscreteUniformDistribution

DiscreteUniformDistribution[{ $i_{\min}, i_{\max}$ }] represents a  
discrete uniform distribution over the integers from  $i_{\min}$  to  $i_{\max}$ . »

Par exemple, une expérience aléatoire consistant à lancer 1 dé, le nombre de points est

```
de = DiscreteUniformDistribution[{1, 6}];
RandomInteger[de, n]
{2, 3, 3, 1, 5, 6, 6, 3, 3, 2, 2, 5, 6, 2, 4, 2, 5, 4,
 4, 5, 5, 5, 1, 5, 2, 5, 4, 5, 6, 3, 4, 1, 5, 4, 4, 6, 1, 3, 4, 1}
```

## ■ Distribution géométrique

?GeometricDistribution

GeometricDistribution[ $p$ ] represents  
a geometric distribution with probability parameter  $p$ . »

Par exemple, une expérience aléatoire consistant à lancer un dé jusqu'à ce que l'on obtienne un six, le nombre de lancers avant d'obtenir le premier six est

```
jets = GeometricDistribution[ $\frac{1}{6}$ ];
RandomInteger[jets, n]
{7, 0, 13, 4, 2, 7, 18, 0, 5, 4, 11, 5, 5, 2, 0, 3, 5, 8,
 6, 7, 6, 7, 0, 7, 3, 6, 0, 0, 23, 2, 1, 1, 27, 0, 4, 4, 6, 1, 0, 1}
```

Si on désire que le dernier jet, qui est un six, soit compté,

```
1 + RandomInteger[jets, n]

{3, 7, 8, 1, 6, 15, 1, 2, 6, 3, 14, 1, 12, 8, 12, 8, 4, 2, 1,
 12, 2, 6, 7, 2, 4, 10, 2, 2, 11, 6, 6, 2, 11, 2, 1, 4, 2, 24, 1, 6}
```

### § 1.3 Nombres pseudo-aléatoires à distribution continue

```
n = 40;
```

#### ■ Distribution uniforme

```
? UniformDistribution
```

UniformDistribution[{*min*, *max*}] represents a continuous uniform statistical distribution giving values between *min* and *max*. »

Par exemple, pour créer un échantillon de *n* mesures principales d'angles, sous l'hypothèse que la distribution est uniforme:

```
angle = UniformDistribution[{-π, π}];
RandomReal[angle, n]

{-3.08028, 0.969671, 2.27415, 0.0602259, 1.85246, -0.586847, -2.48813, 2.69422,
 2.18146, -1.42438, -0.0818245, 0.805137, -0.993308, 2.66945, 2.71752, 1.45863,
 2.02106, 0.242854, 0.540291, -2.72192, -1.14929, -1.0914, -2.34337, -2.33996,
 -2.57387, 0.332647, 1.19147, -0.00588852, -1.20448, -1.9324, 2.57255, -1.64766,
 -2.98082, 2.81899, 2.84292, 0.558487, 0.75726, 0.0423259, 2.31485, 2.18141}
```

#### ■ Distribution normale

```
? NormalDistribution
```

NormalDistribution[ $\mu$ ,  $\sigma$ ] represents a normal (Gaussian) distribution with mean  $\mu$  and standard deviation  $\sigma$ .  
NormalDistribution[] represents a normal distribution with zero mean and unit standard deviation. »

Par exemple, pour créer un échantillon des tailles de *n* personnes, de taille moyenne 1.75 m d'écart-type 0.2 m, sous l'hypothèse que la distribution des tailles est normale:

```
taille = NormalDistribution[1.75, 0.2];
RandomReal[taille, n]

{2.07056, 1.68162, 1.66002, 1.39375, 1.39239, 1.62719, 1.63958, 1.61371,
 2.04569, 2.00756, 1.40057, 1.64944, 1.5993, 2.03974, 1.58805, 1.9175,
 1.67632, 2.06897, 1.89135, 1.78845, 2.00413, 2.07204, 1.75511, 2.03453,
 1.72681, 1.86844, 1.94548, 1.59269, 1.95742, 1.81188, 1.76767, 1.54523,
 1.88773, 1.92284, 1.74084, 2.17602, 1.8296, 1.70421, 1.64808, 1.76311}
```

## § 1.4 Estimation de l'erreur

Pour de grandes valeurs de  $n$ , la moyenne empirique de  $n$  tirages indépendants a une distribution approximativement normale. En notant  $m$  la moyenne empirique,  $s$  l'écart-type empirique et  $\hat{\sigma}_M = \frac{s}{\sqrt{n-1}}$  l'estimation de l'écart-type de la moyenne  $M$ , nous aurons approximativement

la probabilité que l'intervalle  $\left[m - \hat{\sigma}_M, m + \hat{\sigma}_M\right]$  contienne  $\mu$  est de 68.3 % environ;

la probabilité que l'intervalle  $\left[m - 2 \hat{\sigma}_M, m + 2 \hat{\sigma}_M\right]$  contienne  $\mu$  est de 95.5 % environ;

la probabilité que l'intervalle  $\left[m - 3 \hat{\sigma}_M, m + 3 \hat{\sigma}_M\right]$  contienne  $\mu$  est de 99.7 % environ.

### Justification:

Voir *Probabilités II*, § 2.6

## § 1.5 Gestion du temps de calcul

### ■ Recherche de l'efficacité

Pour générer des tirages aléatoires, utilisez si possible les distributions de *Mathematica* et la commande **Random-Array[distr, n]**.

Si la variable est à valeurs  $\{0, 1\}$ , calculez l'écart-type au moyen de la formule  $s = \sqrt{m(1-m)}$ .

Si la variable est discrète et ne comporte qu'un nombre fini de modalités, regroupez les données puis calculez la moyenne et l'écart-type sur les données groupées.

### ■ Contrôle de la durée

La précision s'améliore avec la taille de l'échantillon. Mais la durée des calculs ne peut pas dépasser la limite du supportable, définie par l'utilisateur. Nous adopterons la règle suivante: la taille de l'échantillon est choisie de telle manière que la durée de la simulation est de 1 minute environ.

Pour déterminer le nombre d'expériences, nous effectuons un calcul préalable qui consiste à chronométrer la durée d'un "certain nombre" d'expériences. A partir de là, on calcule combien d'expériences on peut réaliser en 1 minute.

### ■ Exercice 1 - 1

Dans le but d'effectuer la simulation suivante : " Tirer 20 nombres réels pseudo-aléatoires uniformément répartis entre 0 et 1 et afficher ceux qui sont compris entre 0.4 et 0.6", un étudiant a écrit le programme suivant :

```
echant = RandomReal[{0, 1}, 20];

critQ[x_] := 0.4` < RandomReal[{0, 1}] < 0.6`

Select[echant, critQ]

{0.147813, 0.321641}
```

Pourquoi le programme ne donne-t-il pas le résultat escompté ?  
Comment le corriger pour qu'il accomplisse la tâche désirée ?

## ■ Exercice 1 - 2 (facultatif)

Ecrivez un programme qui distribue aléatoirement les 36 cartes du jeu de jass à quatre joueurs.

### Indications

- a) Le début du programme qui construit la liste ordonnée des cartes vous est donné:  
accédez au site <http://www.collegedusud.ch/app/applmaths/>  
suivez les liens **Documents *Mathematica* / Annexes: Simulation**  
et téléchargez le cahier **1-2\_donnees\_exercice.nb**

- b) Pour battre les cartes, on peut utiliser l'idée suivante:  
Charger le module de combinatoire:

```
<< Combinatorica`
```

La fonction **RandomPermutation[k]** produit une permutation aléatoire de la liste  $\{1, 2, 3, \dots, k\}$ . Par exemple,

```
RandomPermutation[10]  
  
{9, 4, 2, 3, 1, 6, 5, 8, 10, 7}
```

Il s'agit d'appliquer la permutation **RandomPermutation[36]** à la liste des cartes pour obtenir un jeu brassé.

- c) Pour le joueur numéro 1, extraire du jeu les cartes numéros 1, 2, 3, 13, 14, 15, 25, 26, 27;  
pour le joueur numéro 2, extraire du jeu les cartes numéros 4, 5, 6, 16, 17, 18, 28, 29, 30;  
etc.

## § 2 Exemples de simulations

### § 2.1 Deux six consécutifs

**Combien de fois faut-il lancer le dé, en moyenne, jusqu'à ce que l'on obtienne 2 six consécutifs?**

Comme question préalable, lançons d'abord le dé jusqu'à l'apparition du premier six.

```
?GeometricDistribution
```

GeometricDistribution[ $p$ ] represents  
a geometric distribution with probability parameter  $p$ . »

```
n6 = GeometricDistribution[ $\frac{1}{6}$ ];
```

Le nombre de lancers d'un dé pour obtenir un six est donné par

```
1 + RandomInteger[n6, 20]
{3, 8, 2, 1, 3, 9, 6, 8, 15, 17, 8, 2, 9, 4, 14, 1, 3, 1, 3, 4}
```

Si, dans la liste précédente, on veut déterminer le nombre de lancers pour obtenir deux six consécutifs (pour la première fois), on peut faire comme suit:

- \* il faut sommer le nombre de coups :  $s = 11 + 3 + 3 + \dots$
- \* il faut interrompre la sommation au premier 1 rencontré, à partir du deuxième rang :
  - le premier terme doit toujours être pris, même si c'est 1;
  - le dernier terme est 1 et fait partie de la somme;
- \* le résultat est  $s = 11 + 3 + 3 + 6 + 1$ .

Le générateur de nombres pseudo-aléatoires qui simule "le nombre de fois qu'il faut lancer le dé pour obtenir deux six consécutifs" est

```
deuxSix[] := Module[{s, t},
  s = 1 + Random[n6]; t = 1 + Random[n6]; s += t;
  While[t > 1, t = 1 + Random[n6]; s += t];
  Return[s]]
```

Mesurons d'abord la durée d'une "petite" simulation:

```
duree = Timing[
  n = 5000;
  x = Table[deuxSix[], {n}]
][[1]]
0.951
```

Accordons maintenant une minute à l'exécution de la simulation principale:

```
n = n * Round[ $\frac{60}{duree}$ ]
315 000
x = Table[deuxSix[], {n}];
```

```

m = N[Mean[x]]

41.9505

StandardDeviationMLE[data_] :=
  StandardDeviation[data] Sqrt[(Length[data] - 1) / Length[data]]

s = StandardDeviationMLE[x];
sm = N[ $\frac{s}{\sqrt{n-1}}$ ]

0.0722621

```

Interprétation : avec une probabilité de 95 %, le nombre cherché  $\mu$  est situé dans l'intervalle

```

{m - 2 sm, m + 2 sm}

{41.806, 42.0951}


```

Ce problème peut être résolu par le calcul des probabilités. Le résultat exact est  $\mu = 42$ .

## § 2.2 Aiguille de Buffon

**On lance une aiguille sur un plancher. La longueur de l'aiguille est 1 et la largeur d'une planche est 2.**

**Quelle est la probabilité que l'aiguille touche 2 planches ?**



### ■ Critère de réussite



Dans un premier temps, organisons une seule expérience.

Notons

$d$  = distance entre le centre de gravité de l'aiguille et la droite la plus proche;

$\theta$  = angle non orienté entre l'aiguille et la perpendiculaire aux droites.





On voit sur les deux graphiques précédents que

l'aiguille coupe un interstice entre deux planches si et seulement si  $\frac{1}{2} \cos(\theta) > d$

```
dist = UniformDistribution[{0, 1}];
```

```
angle = UniformDistribution[{0,  $\frac{\pi}{2}$ }];
```

Mesurons d'abord la durée d'une "petite" simulation:

```
duree = Timing[
  n = 10 000;
  d = RandomReal[dist, n];
  theta = RandomReal[angle, n];
  succes = Count[Table[0.5 Cos[theta[[i]]] > d[[i]], {i, 1, n}], True]
][[1]]
```

0.063

Accordons maintenant une minute à l'exécution de la simulation principale:

```
n = n * Round[ $\frac{60}{duree}$ ]
```

9 520 000

```
d = RandomReal[dist, n];
```

```
theta = RandomReal[angle, n];
```

```
succes = Count[Table[0.5 Cos[theta[[i]]] > d[[i]], {i, 1, n}], True]
```

3 030 788

$$m = N\left[\frac{\text{succes}}{n}\right]$$

0.31836

$$s = \sqrt{m(1-m)};$$

$$sm = N\left[\frac{s}{\sqrt{n-1}}\right]$$

0.00015098

Interprétation : avec une probabilité de 95 %, le nombre cherché est situé dans l'intervalle

$$\{m - 2 sm, m + 2 sm\}$$

{0.318058, 0.318662}


Ce problème peut être résolu par le calcul des probabilités. Le résultat exact est  $\mu = \frac{1}{\pi} \simeq 0.31831$ .

### § 2.3 Calcul d'un volume par la méthode de Monte-Carlo

Considérons le corps  $C = \{(x, y, z) \mid 0 < x < \frac{3}{2}, 0 < y < 1 \text{ et } 0 < z < 4 - x^2 - y^2\}$ .

Quel est son volume ?

`Plot3D[4 - x2 - y2, {x, 0, 1.5}, {y, 0, 1}, PlotRange -> {0, 4}, BoxRatios -> {1, 1, 1}]`



Une expérience élémentaire consiste à placer un point  $(x, y, z)$  au hasard dans le parallélépipède avec

$$\begin{aligned} 0 &\leq x < 1 \\ 0 &\leq y < 1.5 \\ 0 &\leq z < 4 \end{aligned}$$

L'expérience est réussie si le point est situé au-dessous de la surface  $z = f(x, y)$  c'est-à-dire si

$$z < 4 - x^2 - y^2$$

La fréquence de réussite  $m$  est égale au rapport entre le volume "V" cherché et le volume "boite" du parallélépipède:

$$m = \frac{V}{\text{boite}} \quad \Leftrightarrow \quad V = m * \text{boite}$$

```
absc = UniformDistribution[{0, 1.5}];
```

```
ordon = UniformDistribution[{0, 1}];
```

```
cote = UniformDistribution[{0, 4}];
```

Mesurons d'abord la durée d'une "petite" simulation:

```
duree = Timing[
  n = 10 000;
  x = RandomReal[absc, n];
  y = RandomReal[ordon, n];
  z = RandomReal[cote, n];
  succes = Count[Table[z[[i]] < 4 - x[[i]]^2 - y[[i]]^2, {i, 1, n}], True]
][[1]]

0.125
```

Accordons maintenant une minute à l'exécution de la simulation principale:

```
n = n * Round[ $\frac{60}{\text{duree}}$ ]

4 800 000

x = RandomReal[absc, n];
y = RandomReal[ordon, n];
z = RandomReal[cote, n];

succes = Count[Table[z[[i]] < 4 - x[[i]]^2 - y[[i]]^2, {i, 1, n}], True]

3 501 347

m = N[ $\frac{\text{succes}}{n}$ ]

0.729447

s =  $\sqrt{m(1-m)}$ ;
sm = N[ $\frac{s}{\sqrt{n-1}}$ ]

0.000202769

boite = 1 * 1.5 * 4;
vol = m * boite

4.37668

incert = sm * boite

0.00121662
```

Interprétation : avec une probabilité de 95 %, le nombre cherché est situé dans l'intervalle

$$\{\text{vol} - 2 \text{ incert}, \text{vol} + 2 \text{ incert}\}$$

$$\{4.37425, 4.37912\}$$

Ce problème peut être résolu par calcul intégral. Le résultat exact est 4.375.

## Calcul des probabilités exactes

Le calcul des probabilités exactes des exemples des paragraphes 2.1, 2.2 et 2.3 sort des objectifs de ce cours. Les intéressés (en particulier le professeur) peuvent trouver ces calculs dans les annexes:

sur le site <http://www.collegedusud.ch/app/applmaths/>

suivez les liens **Documents Mathematica / Annexes: Simulation**


et téléchargez le cahier **2S-Probabilites\_exactes.nb**

## Exercices du § 2

### ■ Exercice 2 - 1

Perpendiculairement à un vaste grillage, on lance une balle au hasard.

La balle a un rayon de 0.6 et les mailles du filet sont rectangulaires de 4 sur 3.



Quelle est la probabilité que la balle traverse le grillage sans le toucher?

- Calculez la probabilité exacte  $\mu$  de cet événement.  
Indication: utilisez une méthode géométrique).
- Estimez la fréquence de succès  $m$  au moyen d'une simulation.

### ■ Exercice 2 - 2

A l'aide d'une méthode de Monte-Carlo, estimez le volume d'une boule de rayon 1, centrée à l'origine:

$$B = \{(x, y, z) \mid x^2 + y^2 + z^2 \leq 1\}.$$

Commentaire:

$x^2 + y^2 + z^2$  représente le carré de la distance entre le point  $P(x, y, z)$  et l'origine  $O(0, 0, 0)$ :

$$OP^2 = x^2 + y^2 + z^2.$$

$B$  est donc l'ensemble des points  $P$  tels que  $OP^2 \leq 1$ .

### ■ Exercice 2 - 3 (Permutations sans point fixe)

$k$  hommes, avec leurs  $k$  épouses respectives, participent à une soirée dansante. L'animateur de la soirée forme des couples au hasard.

Estimez la probabilité qu'aucun homme ne danse avec son épouse.

Indications:

- 1° Numérotons les couples et représentons la situation par une liste  $v = \{3, 47, \dots, 14\}$  qui signifie, dans le cas particulier,
  - l'homme numéro 1 danse avec la femme numéro 3,
  - l'homme numéro 2 danse avec la femme numéro 47,
  - ...
  - l'homme numéro  $k$  danse avec la femme numéro 14.
- 2° Initialement, la liste est  $v_0 = \{1, 2, \dots, k\}$ .  
Effectuez une permutation au hasard selon la méthode décrite dans l'exercice 1-2 pour battre les cartes.
- 3° Une épreuve est ratée s'il existe un numéro  $j$  tel que  $v[[j]] = j$  c'est-à-dire s'il existe une coïncidence entre les listes  $v$  et  $v_0$ . En *Mathematica*, le test peut s'écrire  

$$\text{If}[\text{MemberQ}[v - v_0, 0], 0, 1]$$
- 4° Il faut organiser une série de  $n$  épreuves.  
Pour de grandes valeurs de  $k$ , l'espérance mathématique  $\mu$  tend vers  $\frac{1}{e} \approx 0.368$

### ■ Exercice 2 - 4

Combien de fois faut-il lancer un dé pour qu'apparaisse l'événement "tirer un six" ?

### ■ Exercice 2 - 5

Combien de fois faut-il lancer trois dés pour que la somme des 3 dés dépasse 15 ?

Indications: structurez votre programme en sous-programmes, chaque sous-programme correspondant à une fonction:

- \* une expérience consiste à lancer 3 dés et à tester si leur somme dépasse 15;
- \* la variable aléatoire est le nombre d'expériences nécessaires pour obtenir une expérience réussie;
- \* une simulation consiste à évaluer un grand nombre de fois la variable aléatoire.

### ■ Exercice 2 - 6

On lance le dé jusqu'à ce que l'on obtienne deux issues consécutives égales puis on considère la somme des points obtenus.

Par exemple :  $1 + 5 + 3 + 5 + 4 + 4 = 22$ .

Estimez la moyenne et l'écart-type des valeurs ainsi obtenues.

Faites un histogramme des résultats obtenus.